SQL Anweisungen effektiv per VBA ausführen
Versionen: Access 2003, 2002/XP und 2000

Durch die direkte Ausführung von SQL-Anweisungen per VBA lassen sich in Access zahlreiche Funktionen vereinfachen und beschleunigen. So können Sie beispielsweise mit einem "DELETE * FROM Tabelle;" schnell und unkompliziert den Inhalt einer temporären Tabelle vor einem Import löschen oder mit "UPDATE Artikel SET VKPreis = VKPreis * 1.05;" in einem Schritt die Preise aller Artikel um 5% erhöhen, ohne mühsam jeden einzelnen Datensatz manuell im Formular anpassen zu müssen. Für die Ausführung solcher SQL-Anweisungen verwenden Sie in der Regel "DoCmd.RunSQL". Alternativ können Sie SQL-Anweisungen aber auch über das "Database"-Objekt oder die DBEngine direkt ausführen.

Beide Methoden haben ihre Vor- und Nachteile und unterscheiden sich je nach Aufgabe in der Ausführungsgeschwindigkeit. Ein Vorteil von "DoCmd.RunSQL" besteht darin, dass Sie beliebige SQL-Anweisungen direkt absetzen können. Ein Nachteil ist jedoch, dass Sie zur Vermeidung der Anzeige von Warnmeldungen immer ein "DoCmd.SetWarnings False" voranstellen müssen. Außerdem werden die per "DoCmd.RunSQL" ausgeführten SQL-Anweisungen standardmäßig mit Transactions ausgeführt, damit die Aktionen ggf. rückgängig gemacht werden können. Das lässt sich zwar durch einen zweiten Parameter mit "True" oder "False" explizit steuern, aber vergisst man diesen Parameter bei SQL-Anweisungen, die keine Transactions benötigen, verzögert sich die Ausführung zum Teil erheblich. Ein weiterer Nachteil besteht darin, dass Sie bei "DoCmd.RunSQL" keine Kontrolle darüber haben, ob durch die Ausführung der SQL-Anweisung Datensätze beispielsweise in einer Tabelle "A" geändert wurden oder nicht. Das kann unter Umständen dazu führen, dass eine nachfolgende Abfrage, die Daten in Tabelle "B" basierend auf der Tabelle "A" aktualisiert, quasi umsonst ausgeführt wird, wenn sich in Tabelle "A" nichts geändert hat.

"Execute" arbeitet generell ohne Transactions und außerdem näher an der Jet Engine, so dass die Ausführung grundsätzlich schneller ist. Zudem werden standardmäßig keine Warnmeldungen angezeigt. Außerdem können Sie eine Eigenschaft "RecordsAffected" abfragen und so feststellen, ob die Ausführung der SQL-Anweisung Datensätze geändert hat oder nicht. Nachteil der "Execute"-Methode ist die zunächst erforderliche Deklaration und Initialisierung entsprechender Objektvariablen sowie die zwingend notwendige Dereferenzierung nach Abschluss der Arbeiten.

Für welche der Methoden Sie sich entscheiden, ist also abhängig von der Aufgabe, die Sie erledigen möchten. So kann man zum einfachen Löschen von Datensätzen aus Tabellen auf den Deklarations- und Initialisierungsaufwand für ein "Execute" verzichten und stattdessen ein simples

DoCmd.RunSQL "DELETE * FROM Tabelle;"

verwenden, während sich für umfangreichere Aktualisierungen der Deklarations- und Initialisierungsaufwand für ein "Execute" durchaus lohnen kann.

Um Ihnen die Unterschiede der verschiedenen Methoden in der Ausführungsgeschwindigkeit vor Augen zu führen, können Sie Test-Funktionen nutzen, die Sie wie folgt einbinden:

1. Wechseln Sie in die VBA-Entwicklungsumgebung und legen Sie ein neues Modul "modSQLTests" an.

2. Kopieren Sie die folgenden Deklarationen und Funktionen in dieses Modul:

Const sqlCreateTable = _
"CREATE TABLE Test (fText text(50), fLong Long);"

Const sqlInsertTextInto = _
"INSERT INTO Test (fText) " & _
"Values('Dies ist ein Testtext...');"

Const sqlInsertLongInto = _
"INSERT INTO Test (fLong) Values(999999);"

Const sqlDropTable = "DROP TABLE Test;"

Public Function TestRunSQL()
  Dim I As Long, J As Long 'Für Schleifen
  Dim sglStart As Single 'Für Zeitmessung

  sglStart = Timer
  DoCmd.SetWarnings False
  For I = 1 To 10
    DoCmd.RunSQL sqlCreateTable, False
    For J = 1 To 10
      DoCmd.RunSQL sqlInsertTextInto, False
      DoCmd.RunSQL sqlInsertLongInto, False
    Next J
    DoCmd.RunSQL sqlDropTable, False
  Next I
  DoCmd.SetWarnings True
  Debug.Print "RunSQL: " & _
              Timer - sglStart & " Sekunden..."

End Function

Public Function TestDBExecute()
  Dim I As Long, J As Long 'Für Schleifen
  Dim sglStart As Single 'Für Zeitmessung
  Dim db As Object 'Für DB-Zugriff

  sglStart = Timer
  Set db = CurrentDb()
  For I = 1 To 10
    db.Execute sqlCreateTable
    For J = 1 To 10
      db.Execute sqlInsertTextInto
      db.Execute sqlInsertLongInto
    Next J
    db.Execute sqlDropTable
  Next I
  db.Close
  Set db = Nothing
  Debug.Print "db.Execute: " & _
              Timer - sglStart & " Sekunden..."

End Function

Public Function TestDBEngineExecute()
  Dim I As Long, J As Long 'Für Schleifen
  Dim sglStart As Single 'Für Zeitmessung
  Dim db As Object 'Für DB-Zugriff
  Dim ws As DAO.Workspace 'Nur Engine-Zugriff

  sglStart = Timer
  Set ws = DBEngine.CreateWorkspace("MyWS", _
             "Admin", "", dbUseJet)
  Set db = ws.OpenDatabase(CurrentDb.Name)
  For I = 1 To 10
    db.Execute sqlCreateTable
    For J = 1 To 10
      db.Execute sqlInsertTextInto
      db.Execute sqlInsertLongInto
    Next J
    db.Execute sqlDropTable
  Next I
  db.Close
  Set db = Nothing
  Set ws = Nothing
  Debug.Print "db.Execute über DBEngine: " & _
              Timer - sglStart & " Sekunden..."

End Function

Jede dieser Funktionen erstellt zehnmal eine neue Tabelle mit einem Text- und einem Zahlenfeld, schreibt neue Datensätze in diese Tabelle und löscht die Tabelle anschließend wieder. Danach wird die benötigte Zeit für die gesamte Ausführung im Direktbereich ausgegeben. Rufen Sie die Funktionen der Reihe nach über den Direktbereich auf:

? TestRunSQL()
? TestDBExecute()
? TestDBEngineExecute()

Sie werden sehen, dass "DoCmd.RunSQL" für diese Aufgabe doppelt soviel Zeit benötigt, wie die "Execute"-Methoden, von denen die Ausführung über das "Database"-Objekt Dank des direkten Zugriffs einen Tick schneller ist, als die Ausführung über "DBEngine", zunächst ein neues Workspace-Objekt eingerichtet und die Datenbank geöffnet werden muss. Das hat allerdings den Vorteil, dass Sie so nahezu genauso schnell auf externe Datenbanken zugreifen können, ohne die aktuelle Datenbank ändern zu müssen.

Mit diesen Testfunktionen können Sie für Ihre Datenbanken die jeweils passende Ausführungsmethode ermitteln, in dem Sie die Konstanten mit den auszuführenden SQL-Anweisungen an Ihre Anforderungen anpassen, die Testfunktionen dann wie gezeigt aufrufen und die Zeiten vergleichen.

